
INTRODUCTION 
 

Au cours de notre cinquième année à l’ISABTP, nous avons suivi une formation théorique sur l’étude 

de problème non linéaire aux éléments finis, en complément du cours dispensé en quatrième année sur 

les problèmes linéaires.  

Afin de mettre à profit les connaissances acquises lors de ce cours, il nous est demandé de modéliser et 

de résoudre un problème non linéaire sur Cast3M. Le logiciel Cast3M, développé par le CEA, permet 

de réaliser une analyse poussée des structures aux éléments finis ; il est notamment très indiqué dans le 

cadre de la résolution de problèmes non linéaires (ex : flambement, fluage, analyses sismiques …).  

Dans cette optique, nous avons choisi de modéliser sur ce logiciel un profilé métallique de type IPE, 

afin d’étudier son comportement vis-à-vis du flambement et du déversement. Le programme réalisé ici 

s’inscrit dans une volonté didactique d’illustrer deux des 3 instabilités vu en cours de charpente 

métallique, afin de permettre une meilleure compréhension des phénomènes physiques qui les 

engendrent. Ainsi, même si dans la cadre de notre projet nous avons fixé des données d’entrée sur le 

type de profilé, et le type d’acier utilisé, le programme se veut adaptable à l’analyse de n’importe quel 

profilé métallique face au flambement et au déversement. 

I. OPTIONS DU MAILLAGE  
 

Les principaux paramètres du maillage sont déclarés à l’aide de l’opérateur « OPTI » (Option) et de 

certains mots-clés.  

I.1. Espace dimensionnel  
 

La dimension de l’espace est déclarée grâce au mot-clé « DIME » (DIMEnsion). Dans le cadre de notre 

projet, nous avons défini un espace dimensionnel 3D « OPTI DIM 3 ». Dans le cas de ce projet la 

modélisation en 3D est primordiale car les instabilités étudiées (flambement et déversement) se 

caractérisent par une déformation hors plan du profilé.  

 

I.2. Types d’éléments  
 

Dans le cadre d’une modélisation aux éléments finis, le maillage constitue le support géométrique de 

notre objet.  C’est cela qui permet de modéliser les phénomènes physiques que nous voulons étudier. Le 

type d’éléments de support de la géométrie est déclaré grâce au mot-clé « ELEM » (ELEMent). Le type 

d’éléments choisi doit être en accord avec la géométrie modélisée, et derrière les calculs qui seront 

effectués aux éléments finis. Comme nous souhaitons étudier une poutre de type IPE en acier face aux 

instabilités de flambement et de déversement, nous avons choisi des éléments de type « TET 4 » qui est 

un élément 3D à quatre nœuds, un tétraèdre. En première approche, cet élément permet de s’adapter aux 

formes de notre structure et notamment dans les zones arrondies du profilé (congés). 

 

I.3. Densité – Taille du maillage  
 

La valeur de la densité correspond à la longueur de référence d’un élément de maillage. Pour définir 

cette donnée, il faut utiliser le mot-clé « DENS » (DENSité). Dans un premier temps, nous avons choisi 

une densité de 0,1.  

P.S : Dans la suite des calculs, cette densité pourra être modifiée afin de raffiner le maillage pour obtenir des 

résultats plus précis.  

I.4. Début du programme  
 

Ainsi, notre début de programme est le suivant :  

OPTION DIMENSION 3 ELEMENT TET4 MODE TRID TRAC OPEN; 

DENSITE 0.10 ; 

II. MODELISATION DU PROFILE 
 

II.1. Opérateurs utilisés pour la modélisation  
 

Nous avons utilisé les opérateurs suivants pour réaliser la modélisation : 

 POIN ou P = x. y. z. : pour créer les points du profilé ; 

 DROI ou D : pour créer les droites ; 

 CERCLE ‘CENT’ : pour créer un cercle à partir de 2 points et du centre ; 

 TRAN : Construit la surface engendrée par la translation d’une ligne suivant un vecteur donné ; 

 ELIM : pour éliminer les éléments communs à deux géométries ; 

 SURF : construction du maillage à l’intérieur du contour précédemment défini ;  

 VOLU : pour générer le volume ; 

 TRAC : pour tracer la géométrie modélisée ; 

 

 

II.2. Choix du profilé  
 

Dans le cadre de notre étude des instabilités « flambement » et « déversement » sur un profilé en acier, 

nous avons décidé de prendre un profilé en forme d’IPE, avec des dimensions non conventionnelles, 

c’est-à-dire non présentes dans les catalogues. Ce choix s’est porté sur le fait d’avoir des déformations 

exagérées sur notre modélisation, afin d’avoir un aspect très « visuel » de ce qu’il se passe lors des 

phénomènes de flambement et de déversement.  

De plus, il est d’ordinaire que les profilés IPE conventionnels soient de classe structurale I ou II. C’est-

à-dire que la limite élastique sera atteinte avant d’observer une instabilité. Dans le cas de ce projet, 

l’étude sera menée avec des méthodes de calcul non linéaire faisant intervenir la plasticité de l’acier. Il 

est donc préférable que les instabilités interviennent avant la plastification de l’IPE étudié. Pour ce faire, 

nous avons allongé l’âme de l’IPE de façon que sa section soit très élancée.  

Nous avons donc commencé par rentrer sur le logiciel Cast3M les dimensions de notre profilé.  

* 
* Données IPE  
* 
h=1.;                                => hauteur du profilé  
b=0.2;                              => largeur des membrures  
tw=0.010;                        => épaisseur de l’âme  
tf=0.016;                          => épaisseur des membrures  
r=0.021;                           => congés 
 
 

II.3. Modélisation du profilé en 3D   
 

II.3.1. Définitions des points  
 

Dans notre modélisation sur Cast3M, nous avons décidé d’exprimer les points par rapport au centre de 

gravité du profilé.  

 

Figure 1 : Points de contour de la section     

 

II.3.2. Définitions des barres et arc-de-cercle  
 

Une fois les points modélisés, nous créons les droites à l’aide de l’opérateur « DROI » et de deux points 

précédemment définis. 

Figure 2 : Contour de la section 

II.3.3. Définition de la surface 
 

Une fois l’ensemble des barres créées, nous pouvons modéliser la surface à l’aide de l’opérateur 

« SURF ».   

 
Figure 3 : Section maillée 

II.3.3. Définition du volume 
 

Afin de créer le volume, nous avons utilisé les opérateurs « VEC » pour engendrer une copie de la 

surface précédemment créée ; et ensuite l’opérateur « VOLU » afin de relier ces deux surfaces en un 

seul objet 3D.   

Figure 4 : Volume maillé 

II.4. Programme Cast3M  
 

 
***********************************   Projet   *************************************** 
 
 
OPTION DIMENSION 3 ELEMENT TET4 MODE TRID TRAC OPEN; 
OPTI EPSI LINEAIRE; 
 
************ Données IPE **************  
 
 
h=1.; 
b=0.2; 
tw=0.010; 
tf=0.016; 
r=0.021; 
 
*********** POINTS PAR RAPPORT CDG DU PROFILE ************ 
 
DENSITE 0.10; 
P1 = (-1*b / 2) 0. (-1*h / 2); 
P2 = (b / 2) 0. (-1*h / 2); 
P3 = (b / 2) 0. ((-1*h / 2)+tf); 
P4 = ((tw / 2)+ r) 0. ((-1*h / 2)+ tf); 
P5 = (tw / 2) 0. ((-1*h / 2)+ tf + r); 
P6 = (tw / 2) 0. ((h / 2) - tf - r); 
P7 = ((tw / 2) + r) 0. ((h / 2) - tf); 
P8 = (b / 2) 0. ((h / 2) - tf); 
P9 = (b / 2) 0. (h / 2); 
P10 = (-1*b / 2) 0. (h / 2); 
P11 = (-1*b / 2) 0. ((h / 2) - tf); 
P12 = ((-1*tw / 2) - r) 0. ((h / 2) - tf); 
P13 = (-1*tw / 2) 0. ( (h / 2) - tf - r); 

P14 = (-1*tw / 2) 0. ((-1*h / 2) + tf + r); 
P15 = ((-1*tw / 2) - r) 0. ((-1*h / 2) + tf); 
P16 = (-1*b / 2) 0. ((-1*h / 2) + tf); 
CE1= ((tw / 2) + r) 0. ((-1*h / 2) + tf + r);  
CE2 = ((tw /2) + r) 0. ((h / 2 )- tf - r); 
CE3= ((-1*tw / 2) - r) 0. ((h / 2) - tf - r);  
CE4 = ((-1*tw /2) - r) 0. ((-1*h / 2 )+ tf + r); 
 
 
PTOT = P1 ET P2 ET P3 ET P4 ET P5 ET P6 ET P7 ET P8 ET P9 ET P10 ET P11 ET P12 ET P13 ET 
P14 ET P15 ET P16 ET CE1 ET CE2 ET CE3 ET CE4; 
TRACER PTOT; 
 
 
************** BARRES *************** 
 
B12= DROI P1 P2; 
B23= DROI P2 P3; 
B34= DROI P3 P4; 
C45= CERCLE 'CENTR' P4 CE1 P5; 
B56= DROI P5 P6; 
C67= CERCLE 'CENTR' P6 CE2 P7; 
B78= DROI P7 P8; 
B89= DROI P8 P9; 
B910= DROI P9 P10; 
B1011= DROI P10 P11; 
B1112 = DROI P11 P12; 
C1213 = CERCLE 'CENTR' P12 CE3 P13;  
B1314= DROI P13 P14; 
C1415 = CERCLE 'CENTR' P14 CE4 P15; 
B1516= DROI P15 P16; 
B161= DROI P16 P1; 
 
TRACER (B12 ET B23 ET B34 ET C45 ET B56 ET C67 ET B78 ET B89 ET B910 ET B1011 ET B1112 
ET C1213 ET B1314 ET C1415 ET B1516 ET B161); 
CO= B12 ET B23 ET B34 ET C45 ET B56 ET C67 ET B78 ET B89 ET B910 ET B1011 ET B1112 ET 
C1213 ET B1314 ET C1415 ET B1516 ET B161; 
SURF1 = SURF 'PLAN' CO; 
TRACER SURF1; 
 
***************** CREATION DU VOLUME *************** 
 
VEC1 = 0. 10. 0.; 
V0 = SURF1 VOLU 100 'TRAN' VEC1 ; 
B12_2=B12 PLUS VEC1; 
ELIM V0 B12_2; 
TRACER V0; 

III. Caractéristiques mécaniques du modèle 
 

III.1 Sollicitations 
 

La poutre sera soumise à un effort uniformément réparti sur toute sa longueur. Cet effort sera appliqué 

sur la surface de la semelle supérieure. Etant donné que la génération du volume s’est fait par translation 

de la section, les points et surfaces créés ne sont pas nommés. Il faut donc réussir à extraire du volume 

la surface de la semelle supérieure pour la nommer afin d’y appliquer la charge. 

Pour ce faire nous avons récupéré la surface supérieure avec le codage suivant : 

SEXT=V0 FACE 3; 

CH2=MANU CHML SEXT SCAl 1; 

Z=COORD 3 CH2; 

Fsup=Z ELEM MAXI STRI; 

 

Ce bout de code permet de sélectionner tous les points maximaux de la face 3 du volume (normal à l’axe 

3). On peut vérifier ce programme en traçant la surface « Fsup ». 

 

Figure 5 : Surface supérieure pour application du chargement 

 

Maintenant que la surface est nommée nous pouvons appliquer la charge uniformément répartie sur la 

semelle. Il faut ici faire attention car cette charge est appliquée sur une surface, il s’agit donc d’une 

charge surfacique. Pour la traduire en une charge linéique il suffit de multiplier l’intensité de la charge 

par la largeur de la semelle supérieure. 

La charge ici appliquée est dirigée selon l’axe Z avec le mot « MASS ». Son intensité est de 100000 Pa 

ce qui représente une charge linéique de 20 kN/m.  

QY1=100000; 

F1 = PRESSION 'MASS' MOD1 Fsup QY1; 

 

III.2 Conditions aux limites 
 

Le problème étudier sera une poutre isostatique sur deux appuis. Pour modéliser ce schéma mécanique, 

nous avons choisi d’appuyer la poutre sur les deux extrémités des semelles inférieures. Nous avons donc 

bloqué les déplacements en UZ mais aussi en UY et UX pour rendre la matrice de géométrie inversible. 

Sans le blocage en UY et UX la poutre serait instable. Afin de modéliser des appuis simples nous avons 

laissé libre les rotations sur ces deux barres selon l’axe RY.  

 

CL1 = BLOQUE B12 UX UY UZ; 

CL2 = BLOQUE B12_2 UX UY UZ; 

CLTOT = CL1 ET CL2; 

 

Note : B12_2 correspond à la barre B12 translaté de la longueur de la poutre (VEC1). 

 
 

IV. Méthodes de calculs non-linéaires  
 

Pour étudier les phénomènes d’instabilité d’un profilé d’acier nous étudierons deux méthodes de calcul 

disponibles sur Cast3M. L’une utilise une procédure définie (Flambage) et l’autre utilise une procédure 

pas-à-pas. Pour que l’étude soit cohérente nous utiliserons la même géométrie, les mêmes conditions 

aux limites, le même maillage ainsi que les mêmes sollicitations. 

IV.1. Méthode de bifurcation 
 

Cette méthode permet de calculer le flambement et le déversement élastique d’une géométrie. Elle utilise 

la théorie d’Euler. Le principe réside à trouver deux états d’équilibre limite. L’un quand l’élément est 

rectiligne et la force axiale. Alors on trouve un équilibre stable sous des faibles charges (1). Une charge 

limite d’équilibre stable peut être trouvée quand une perturbation ne permet pas à la section de revenir 

dans son état initial. L’autre état d’équilibre correspond à un élément très faiblement courbé pour lequel 

une faible augmentation de charge conduit à des grands déplacements et une impossibilité de retrouver 

les déplacements initiaux (2).  

Figure 6 : Schéma explicatif méthode de bifurcation 

 

Cette méthode est facile à mettre en place car il n’y a pas besoin de créer une courbure initiale de la 

poutre.   

Cependant pour utiliser la fonction flambage il faut préalablement résoudre le problème élastique pour 

connaitre les contraintes dans la géométrie. Pour cela nous entrons les caractéristiques de l’acier ainsi 

que le modèle mécanique à étudier. 

********* Caractéristiques du matériau ********* 

YG1=210.E9; 

NU1=0.3; 

********* définition du modèle de calcul ******** 

MOD1 = MODELE V0 MECANIQUE ELASTIQUE; 

MAT1= MATE MOD1 'YOUN' YG1 'NU' NU1; 

 

La résolution du problème élastique se fait classiquement soit : 

******** RESOLUTION PROBLEME ELASTIQUE ******** 

RIG1 = RIGI MOD1 MAT1; 

RITOT = RIG1 ET CLTOT; 

 

DEP1 = RESOU RITOT F1; 

SIG1 = SIGM 'LINE' MOD1 MAT1 DEP1 ; 

 

Les résultats des contraintes dans le maillage sont stockés dans « SIG1 ». Le mot « line » permet de ne 

sélectionner que les termes de premier ordre. 

Pour utiliser l’opérateur flambage nous devons créer une table dans laquelle nous placerons toutes les 

informations nécessaires au calcul. On retrouve alors dans cette table les éléments suivants : 

- Le modèle, indiqué par le mot clé « OBJM » : MOD1 

- Le coefficient multiplicateur minimal de la sollicitation, indiqué par le mot-clé « LAM1 » : 

0.001 

- Le coefficient multiplicateur maximum de la sollicitation, indiqué par le mot-clé « LAM2 » : 

100 

- Le nombre de mode d’instabilité étudié, indiqué par le mot-clé « NMOD » : 2 

- Les conditions limites du modèle, indiquées par le mot-clé « CLIM » : CLTOT 

- Les caractéristiques du matériau, indiquées par le mot-clé « MATE » : MAT1 

- Les contraintes issues du calcul élastiques, indiquées par le mot-clé « SIG1 » : SIG1 

Cette table peut être implémentée par d’autres informations. 

Le code se traduit par ce qui suit : 

TAB2 = TABLE; 

TAB2.OBJM = MOD1; 

TAB2.LAM1 = 0.001; 

TAB2.LAM2 = 100.; 

TAB2.NMOD = 2; 

TAB2.CLIM= CLTOT; 

TAB2.MATE=MAT1; 

TAB2.'SIG1'=SIG1; 

 

On peut ensuite lancer le calcul avec le code suivant :  

TAB1 = FLAMBAGE TAB2 ; 

Les résultats seront stockés dans la table TAB1. On peut notamment retrouver le coefficient 

multiplicateur de la charge pour atteindre un mode propre, les déformations, … 

Figure 7 : Table après résultats de la fonction FLAMBAGE 

 

On peut par exemple tracer la déformée de la structure au mode d’instabilité 1 : 

def_Mod1 = defo V0 TAB1 . 1 . depl; 

trac def_Mod1 cach; 

trac def_Mod1 TAB1 . 1 . depl cach; 

 

 
Figure 8 : Déplacement au mode 1 

Notons ici que le déplacement indiqué est unitaire. L’image ci-dessus nous montre donc le déplacement 

du profilé lors de son déversement.  

Figure 9 : Contraintes de Von-Mises au mode 1 

 

On peut faire de même avec la déformée de la structure au mode d’instabilité 2 : 

def_Mod2 = defo V0 TAB1 . 2 . depl; 

trac def_Mod2 cach; 

trac def_Mod2 TAB1 . 2 . depl cach; 

 

 
Figure 10 : Déplacement au mode 2 

Les principales informations issues de cette étude sont les coefficients multiplicateurs de la charge. Avec 

ces coefficients on peut connaitre la charge limite pouvant être appliquée sur le profilé. C’est cette charge 

que l’on pourra comparer avec l’étude de déversement par déviation.  

 

 
Figure 11 : Coefficient multiplicateur selon le mode d'instabilité 

 

Dans ce cas nous trouvons un coefficient multiplicateur pour le premier mode de 3,4102. C’est-à-dire 

que la charge appliquée sur le profilé lors du déversement est de 20 KN/m * 3,4102 = 68,204 KN/m. Le 

moment fléchissant mi-travée est donc de 
଼.ଶ × ଵమ

଼
=  852,5

ே


.  

 

IV.2. Méthode de divergence  
 

La méthode de divergence consiste à étudier pas à pas différents équilibres statiques du profilé étudié 

en prenant aussi en compte la plastification possible. A cause d’une courbure initiale du profilé, les 

efforts extérieurs sont amplifiés et contribuent ainsi à l’accroissement des contraintes jusqu’à la 

plastification. On observe une divergence quand il y a une forte augmentation de la charge et des 

déplacements, ne permettant plus l’équilibre de la section. On dit alors qu’il y a instabilité par divergence 

d’équilibre. Ainsi on peut voir, sur le diagramme de force-déplacement ci-dessous, que le comportement 

linéaire est représenté en (1). On observe ensuite l’asymptote représentant les déplacements modérés de 

la théorie du second ordre. C’est cette asymptote qui donne une estimation de la force critique. Ensuite 

on retrouve une augmentation de la force exponentielle représentant la non linéarité en grand 

déplacement.  

Figure 12 : Schéma explicatif méthode de divergence 

 

Dans ce chapitre nous nous proposons donc d’étudier le profilé initialement courbé en augmentant petit 

à petit la charge extérieure afin d’arriver à la première instabilité.  

 

Pour générer une courbure sur l’ensemble du profilé nous utilisons le champ de déplacement trouvé avec 

la procédure « flambage » au chapitre précèdent. Nous estimerons aussi, en première approximation, 

que la charge critique est celle obtenue pour le premier mode d’instabilité (par bifurcation).  

 

Pour courber le volume initial (V0) on crée un champ de déplacement ch1 correspondant à 0,1 millimètre 

du déplacement obtenu par bifurcation au mode 1. Avec la fonction « plus » le volume V0 va s’aligner 

sur le champ ch1. De ce fait nous sommes certains d’avoir une courbure initiale qui pourra engendrée 

une instabilité de premier mode.   

 

On utilise la syntaxe suivante :  

ch1 = 0.0001* TAB1 . 1 . depl; 

deplacer V0 plus ch1; 

TRAC V0; 

On obtient bien un volume courbé :  

 

 
Figure 13 : Volume avec courbure initiale 

 
Pour réaliser la méthode de divergence nous avons besoin de passer le modèle en plastique parfait. Nous 

devons donc choisir une limite élastique (240 MPa) et un module de poisson (0,3).  

Nous utilisons la syntaxe suivante : 

 

SIGY1=240.E6; 

YG1=210.E9; 

NU1=0.3; 

 

MOD1 = MODELE V0 MECANIQUE ELASTIQUE PLASTIQUE PARFAIT; 

MAT1= MATE MOD1 'YOUN' YG1 'NU' NU1 'SIGY' SIGY1; 

 

Dans une étude normale non linéaire (en plasticité) nous devrions résoudre le problème élastique avant 

pour connaitre la charge qui solliciterait la section hors de son domaine élastique. Cependant grâce à 

l’étude de bifurcation nous avons déjà une approximation de la charge limite avant déversement. On se 

servira donc du coefficient multiplicateur de mode de déversement 1 obtenue au chapitre III.1 soit 

3,4102. On aura donc une charge limite de 3,4102*100000=341020 Pa. 

 

QCRIT=(TAB1 . 1 . LAMB)*QY1; 

 



 

 

Etude d’instabilité en plasticité d’un profilé en acier 
Méthodes de bifurcation et de divergence  

18 MARTINEZ Rodolphe – PEREIRA DE SOUZA Laetitia  

Pour utiliser la procédure pas-à-pas nous devons créer une charge qui augmente au cours du temps. 

Comme précédemment la charge est appliquée sur la surface supérieure. Pour être sûr d’entrer dans le 

domaine plastique nous prendrons une charge maximale 3 fois supérieure à la charge critique Qcrit. On 

crée ensuite deux listes de réel correspondant respectivement aux temps et au chargement. Par 

simplification ces listes sont de même taille. C’est-à-dire que pour un pas de temps franchi, la même 

unité de chargement est appliquée (Exemple : A temps = 80, le chargement appliqué est de 80 Pa). 

On crée ensuite une évolution linéaire de la force appliquée en fonction du temps. On précise aussi que 

le chargement est de type mécanique. On utilise la syntaxe suivante : 

************** Création de la charge variant avec le temps ***************** 

*** f(t) = F1 * lambda(t) ; avec F1 unitaire égal à 1 

 

F1 = PRESSION 'MASS' MOD1 Fsup 1.;  

 

*********** Evolution du multiplicateur en fonction du pseudo temps *************** 

 

QMAX = 3.*QCRIT ; 

TEMPS_1 = prog 0. QMAX; 

FORCE_1 = prog 0. QMAX; 

EVOL1 = EVOL MANU TEMPS_1 FORCE_1; 

 

CHAR1 = CHARGEMENT 'MECA' F1 EVOL1 ; 

 

On peut ensuite implémenter la table pour réaliser le calcul non linéaire (pas-à-pas) :  

On renseigne les caractéristiques principales du problème soit : 

- Le chargement ; 

- Les données du modèle ; 

- Les caractéristiques matériaux ; 

- Les conditions aux limites (similaires à celles utilisées dans l’étude de bifurcation) ; 

- L’activation de la prise en compte des grands déplacements et la méthode du lagrangien 

actualisé ; 

- Les temps de calculs. Le temps débute à 0. Il avance jusqu’à 0.7* Qcrit par pas de dixième de 

Qcrit. Ensuite il avance par pas de centième de Qcrit jusqu’à 1,1*Qcrit. Ce raffinement autour 

de Qcrit permet de se rapprocher de la charge critique du modèle par divergence sans qu’elle 

soit atteinte entre deux pas. Nous aurons donc un résultat plus précis. 

La synthèse utilisée est la suivante : 

**************Procédure Pas-a-pas*************** 

 

TAB1 = TABLE 'PASAPAS'; 

TAB1.CHARGEMENT=CHAR1; 

TAB1.MODELE=MOD1; 

TAB1.CARACTERISTIQUES=MAT1; 

TAB1.BLOCAGES_MECANIQUES=CLTOT; 

TAB1.grands_deplacements = VRAI; 

TAB1.UPDATE_LAGRANGIAN = VRAI; 

TAB1.TEMPS_CALCULES=PROG 0. pas (QCRIT/10.)  (0.7*QCRIT) PAS (QCRIT/1000.) 

(1.1*QCRIT); 

 

PASAPAS TAB1; 

 

On obtient une table résumant les résultats. On peut alors extraire le déplacement et la déformation sous 

la charge de déversement. On obtient alors : 

 

 
Figure 14 : Déplacement au dernier pas de temps 

Le déplacement maximum est d’environ 1 cm sur la semelle supérieure du profilé. 

 

On peut faire de même pour les contraintes : 
 

 

Figure 15 : Contraintes au dernier pas de temps 

La contrainte maximale est d’environ 290 MPa. Le profilé est donc plastifié au niveau de ces appuis. 

Cette plastification localisée au niveau des semelles inférieures (aux extrémités) est due aux conditions 

limites saisies pour les appuis. En effet, les extrémités des semelles sont seulement libres en rotation 

autour de leur axe (y-y). La plastification provient probablement du déplacement hors plan de l’âme par 

rapport aux semelles inférieures. 

Pour connaitre le coefficient multiplicateur de la charge qui mène jusqu’à l’instabilité nous traçons une 

évolution de la charge en fonction du temps. Nous superposons aussi le coefficient multiplicateur de 

charge trouvé avec la méthode de bifurcation.  

La syntaxe utilisée est la suivante : 

PROGD=PROG; 

PROGF=PROG; 

PROGFC=PROG; 

REPETER BOU1 N1; 

  D1=MAXI TAB1 . DEPLACEMENTS . (&BOU1 - 1) AVEC (MOTS UX) ABS; 

  F1=TAB1 . TEMPS . (&BOU1 - 1); 

  PROGD=PROGD ET D1; 

  PROGF=PROGF ET F1; 

  PROGFC = PROGFC ET QCRIT; 

FIN BOU1; 

EVOL1 =EVOL MANU PROGD PROGF; 

EVOL2 = EVOL MANU PROGD PROGFC; 

dess ( evol1 ET evol2); 

 

 
Figure 16 : Chargement en fonction du déplacement Ux, bifurcation et divergence 

 

Nous remarquons que la courbe est non linéaire et qu’elle semble avoir une asymptote aux environs de 

3,32*10^5 MPa. La différence entre les coefficients multiplicateur obtenues entre les deux méthodes est 

de 
ଷ,ସଵିଷ,ଷଶ

ଷ,ସଵ
× 100 = 2,64 % . Cette faible différence montre la cohérence des études menées. Le 

comportement observé semble adéquat avec le principe de l’étude par divergence.  

Cependant, avant d’obtenir de tels résultats, nous devons faire attention à plusieurs paramètres de calcul. 

Le premier d’entre eux est la déformation initiale donnée au profilé. Si celle-ci est trop grande il se 

pourrait que le déversement ait lieu dès les premiers pas de temps. Nous avons donc une courbe peu 

exploitable que donne des résultats très approximatifs. Par exemple, le tracé ci-dessous a été réalisé avec 

un déplacement initial de 10 cm.  

 

Figure 17 : Chargement en fonction du déplacement Ux, bifurcation et divergence, avec grande déformation initiale 

 

Le résultat de la méthode de divergence représente environ 50% de celui de la méthode par bifurcation. 

Cet écart n’est pas acceptable. On remarque bien que la courbe se stoppe avant d’atteindre une 

asymptote. Comme dit précédemment, ce problème est dû au fait que le déversement intervienne bien 

trop tôt. Cela introduit alors un autre point sur lequel nous devons porter une attention particulière lors 

de la rédaction du code, la taille des pas de temps. En effet ce paramètre est réglé dans la ligne suivante : 

TAB1.TEMPS_CALCULES=PROG 0. pas (QCRIT/10.)  (0.7*QCRIT) PAS (QCRIT/1000.) 

(1.2*QCRIT); 

Cela veut dire que le temps débute à 0 et va jusqu’à 0.7*Qcrit= 2,387 par pas de Qcrit/10 soit 0,341. La 

courbe s’interrompt à environ 1,7. Avec le déplacement initial de 10 cm et une augmentation brutale de 

0,341 (en coefficient multiplicateur) il est probable que le programme n’arrive pas à trouver d’équilibre 

entre la section plastifiée et les forces extérieures. Il est donc important d’adapter le pas de temps. Ainsi 

la diminution de pas de temps entre 0,7*Qcrit et 1,2Qcrit permet de générer une asymptote avant de ne 

plus atteindre l’équilibre. Cette asymptote et bien visible en prenant un déplacement de 0,1 mm. 

Lançons maintenant un calcul plus raffiné avec une déformation initiale de 1 * 10-3 mm et un pas de 

temps de Qcrit/10000 soit 3,41*10-4.  

 

Figure 18 : Chargement en fonction du déplacement Ux, bifurcation et divergence, modèle affiné 

 

Ce calcul est beaucoup plus long que les deux derniers, environ 4 minutes, or le coefficient 

multiplicateur à l’asymptote reste égal à 3,32. Dans ce cas le raffinement du modèle de calcul ne nous a 

pas permis d’améliorer la qualité des résultats. Cependant grâce à ce calcul nous savons que les 

caractéristiques introduites lors du premier calcul (déformation de 0,1 mm et pas de temps à Qcrit/1000) 

permettaient d’avoir un modèle convergent.  

 

 

V. Conclusion de l’étude Cast3m 
 

L’étude Cast3m nous a permis d’étudier le déversement d’un profilé d’acier sollicité par une charge 

uniformément répartie sur la semelle supérieure. Pour ce faire deux méthodes de calcul ont été utilisées. 

La première, dite par bifurcation, fait appel à la fonction « flambage » du code Cast3m. Elle permet 

d’obtenir rapidement les coefficients multiplicateurs de la charge initiale déclenchant les différents 

modes propres. Avec cette méthode nous avons conclu que le coefficient multiplicateur de la charge au 

premier mode de déversement est de 3.4, soit un moment fléchissant maximum de 852,5 kN.m. D’autre 

part, nous avons utilisé la méthode de divergence qui fait appel à la fonction « PAS-A-PAS ». Cette 

méthode nécessite plus de programmation car une déformation initiale ainsi qu’une évolution de la 

charge doivent être introduites. Avec cette méthode nous avons obtenu un coefficient multiplicateur de 

3,32 soit un moment maximum de 830 kN.m.  

On peut conclure que le moment limite avant le premier mode de déversement est de 
଼ହଶ.ହା଼ଷ  

ଶ
=

841.3 𝑘𝑁.  

 

VI. Comparaison avec les méthodes de RDM 
 

En RDM des formules permettent d’obtenir le moment critique de déversement élastique d’un profilé 

en fonction des conditions d’appuis. On peut ainsi réaliser un calcul et le comparer aux résultats 

précédemment obtenus. Etant donné que l’étude de Cast3m a été menée en plasticité, le moment critique 

de déversement élastique devrait logiquement être plus faible.  

La formule générale utilisée est décrite ci-dessous : 

 

Figure 19 : Equation du moment critique de déversement 

 

C1, C2 sont des coeficients dépendant des conditions de chargement et des appuis d’extrémités. Ils sont 

respectivement égaux à 0,97 et 0,30. 

kz et kw sont des coefficient liés aux conditions d’appuis. Dans notre cas, la semelle inférieure empèche 

des rotations dans l’axe de l’âme et de la portée du profilé. Alors kz = kw = 0,5. 

Avec les caractéristiques géométrique de la section on trouve : 

𝐼௭ = 2.14 × 10ିହ 𝑚ସ 

𝐼௧ = 8.69 ×  10ି 𝑚ସ 

𝐼௪ = 5.18 × 10ି 𝑚ସ 

𝐺 = 80.67 𝑀𝑃𝑎 

Avec ces données on trouve 𝑀 = 690,8 𝑘𝑁. 𝑚 

On observe donc une différence de résultat de 
଼ସଵ.ଷିଽ.ଷ

଼ସଵ.ଷ
= 18% qui s’explique par le fait que le l’acier 

du profilé entre dans le domaine plastique dans les études de bifurcation et divergence.  

Conclusion 
 

Au cours de ce projet, nous avons utilisé Cast3m pour réaliser et comprendre deux façons d’étudier un 

effet de second ordre sur un profilé tout en entrant dans le domaine plastique de l’acier. L’effet de second 

ordre étudié est le déversement d’un IPE soumit à une charge répartie. Nous avons réussi à trouver des 

résultats cohérents et du même ordre de grandeur avec ces deux méthodes. Nous avons aussi compris 

l’enjeux du réglage des paramètre initiaux des calculs non linéaires (procédure pas-à-pas) ; notamment 

la précision de la courbure initiale et des pas de temps pour l’étude par divergence. Enfin, un simple 

calcul de résistance des matériaux dans le domaine élastique de l’acier nous a permis de valider l’ordre 

de grandeur de nos résultats et l’intérêt d’un calcul en plasticité.  

Nous conclurons en affirmant que la prise en compte des non-linéarités des matériaux apporte une vraie 

valeur ajoutée dans les calculs de second ordre. Un code comme celui présenté dans ce projet pourrait 

servir à un bureau d’étude pour résoudre des problèmes simples en prenant en compte différents 

comportement de matériaux comme les paliers plastiques inclinés ou l’écrouissage.  

 

 

 

 

 

 


